

Designing With Op Amps

Designing your own circuits with BASIC.

DERRICK RENAUD

Operational Amplifiers have two inputs and one output (in their most common form). With just a few connections they can perform many different functions, such as Amplifiers, Integrators, Differentiators, Filters, etc. This article will be limited to using them as amplifiers only. For anyone wanting to go in depth, I would suggest reading the *Master Op Amp Applications Handbook*, written by Harry W. Fox and published by TAB Books.

The Inverting Amplifier

The Inverting Amp is shown in Figure 1. Its function is to produce an inverted copy of the input at the output with a specified amount of gain. The gain is determined by the ratio of R2 to R1 and can be determined by the formula: (Voltage) GAIN = $-R2/R1$.

R2 is a feedback resistor used for reducing the op amp's maximum gain which also reduces distortion. Higher values of R2 produce more Offset voltage at the output. So R2 should be kept as low as possible (offset voltage is the output's variance from ground, when the input is grounded.)

R1 is the Input resistance of the circuit. It should be determined first, if it is critical to your circuit application.


R3 is used to reduce the offset voltage. It's optional, if you are not worried about offset; just replace it with a short (0 ohms) to ground if it is not needed.

C is used to block DC which would otherwise be amplified. If a DC amp is required then it can be replaced with a short.


The Non-Inverting Amplifier

The Non-Inverting Amp is shown in Figure 2. Its function is to produce an output of the same polarity as the input, with a specified amount of gain. The gain is determined by the ratio of R2 to R1 and can be determined by the formula: (Voltage) GAIN = $R2/R1 + 1$. R2's function is the same as in the Inverting Amplifier.

R3 is used to reduce the Offset voltage. Since a non-inverting amp without R3



Inverting Amplifier as drawn by the BASIC program.



Non-Inverting Amplifier as drawn by the BASIC program.

will have a very high input resistance, then R3 in this circuit is the input resistance.

C is used to block DC from being amplified. R3 has to be used when C is used to provide a bias path to ground for the + input bias current. (The - input is biased by R2 in both the inverting and non-inverting Amplifiers.) In the case of a DC amplifier, C can be replaced with a short. R3 can be removed because the DC source will provide the bias path.

Other Parameters

Once the gain has been determined, you can begin to worry about Slew Rate, Band Width Gain and Frequency compensation.

Slew rate (SR) determines how fast a voltage can rise or fall. It is determined by the maximum frequency and the peak to peak output voltage, using the formula: $SR = 6.28 \times F_{max} \times V_{pp} \times 10^{-6}$. A value greater than this should be used to give the amplifier extra headroom.

Band Width Gain (BWG) determines the maximum amount of gain a certain frequency bandwidth can have. It can be determined by the formula: $BWG = (\text{Voltage}) \text{Gain} \times F_{max}$. In practice a value

at least 10 times this is used to make certain there will be enough BWG.

Slew Rate and Band Width Gain, once calculated, are used to find the proper op amp for the job; just check them against values in different data books.

Frequency compensation is required to reduce oscillations caused by internal phase shifts in the op amp. The simplest way to do this is called Single Pole Compensation, for which the manufacturer supplies pin connections. All that is required is a single capacitor determined by the formula: $C_{comp} = (C_s \times R1) / (R1 + R2)$; where C_s is usually 30 pF. C_{comp} is in pF.

Numbers, Numbers, Numbers

If all this seems like a lot of paperwork to get the values needed, think again. All the values are calculated by given mathematical formulas.

Give a computer the formulas and presto. One such program is shown in Listing 1.

About the Program

The program was written in Microsoft Basic using an APPLE IIe running CP/M. When it is run, it will ask you to input five parameters. The only parameter that has to be entered is the gain. If the input resistance is not given then a suitable value will be determined. The other parameters are only needed to calculate the DC blocking capacitor, slew rate and band width gain.

The program displays a schematic diagram of the chosen amp along with all values. It will choose common resistor and capacitor values for you. It's very easy to use - just type it in, save it, and start punching in values.

Subroutine 2570 - is used to clear the screen and can be replaced by whatever your computer Clear Screen command is.

A Final Note

Amp stages can be cascaded for higher gains.

$$\text{Voltage Gain} = VG1 \times VG2 \times \dots$$

$$\text{Gain (dB)} = G1 + G2 + \dots$$

```

1000 DIM Z(168), Y(33)
1010 GOSUB 2580
1020 PRINT:PRINT:PRINT "CALCULATING RESISTOR TABLE... Please wait"
1030 REM *** LOAD RESISTOR TABLE ***
1040 FOR A=1 TO 24
1050 READ B
1060 FOR C=0 TO 6 : Z(A+24*C)=B*10^C : NEXT C
1070 NEXT A
1080 REM *** LOAD CAPACITOR TABLE ***
1090 PRINT : PRINT "CALCULATING CAPACITOR TABLE... Please wait"
1100 FOR A=1 TO 11
1110 READ B
1120 FOR C=0 TO 2: Y(A+11*C)=B*10^C : NEXT C
1130 NEXT A
1140 GOSUB 2580
1150 PRINT "          *****"
1160 PRINT "          #   AMPLIFIER   DESIGN   #"
1170 PRINT "          #           (c) 1988     #"
1180 PRINT "          #   BY Derrick Renaud   #"
1190 PRINT "          *****"
1200 PRINT : PRINT : PRINT
1210 PRINT " Do you need: 1. Inverting Amplifier"
1220 PRINT "           2. Non-Inverting Amplifier"
1230 PRINT
1240 INPUT " TYPE 1 or 2 ";T : IF T<>1 AND T<>2 THEN 1240
1250 REM *** CLEAR PARAMETER VALUES ***
1260 F1=0: F2=0: VM=0: G=0: RI=0
1270 REM *** INPUT DATA ***
1280 GOSUB 2580 : PRINT : PRINT : PRINT SPC(20);
1290 IF T=2 THEN PRINT "Non-";
1300 PRINT "Inverting Amplifier Parameter Input"
1310 PRINT : PRINT
1320 PRINT "NOTE: IF Fmin = 0 THEN C WILL NOT BE CALCULATED."
1330 PRINT "      IF Fmax = 0 THEN SR AND BWG WILL NOT BE CALCULATED."
1340 PRINT "      IF Vmax = 0 THEN SR WILL NOT BE CALCULATED."
1350 PRINT "      IF Ri = 0 THEN IT IS ASSUMED THAT IT IS NOT CRITICAL AND A REASONABLE"
1360 PRINT "              VALUE WILL BE DETERMINED."
1370 PRINT "      PRESSING 'RETURN' WILL DEFAULT A VALUE OF 0."
1380 PRINT : PRINT
1390 INPUT "LOW FREQUENCY LIMIT IN Hz (Fmin) ";F1 : IF F1<0 THEN 1390
1400 INPUT "HIGH FREQUENCY LIMIT IN Hz (Fmax) ";F2 : IF F2=0 THEN 1430
1410 IF F2<F1 THEN 1400

```

Designing With Op Amps

```
1420 INPUT "MAXIMUM RMS OUTPUT VOLTAGE (Vmax) ";VM : VM=ABS(VM)
1430 INPUT "GAIN IN dB ";G : IF G=0 AND T=2 THEN 1430
1440 INPUT "INPUT IMPEDANCE IN OHMS (Ri) ";RI : IF RI<0 THEN 1440
1450 PRINT : PRINT : PRINT "IS ALL DATA CORRECT? ";
1460 A$=INKEY$ : IF A$="N" THEN 1280
1470 IF A$>"Y" THEN 1460
1480 V=10^(G/20)
1490 REM *** CHECK INPUT IMPEDANCE ***
1500 RX=RI : IF RI<>0 THEN 1610
1510 R2=10000
1520 IF G>15 THEN R2=47000!
1530 IF G>30 THEN R2=100000!
1540 IF G>45 THEN R2=470000!
1550 IF T=1 THEN R1=R2/V ELSE R1=R2/(V-1)
1560 RT=R1 : GOSUB 2290 : R1=RT
1570 R3=R1*R2/(R1+R2)
1580 RT=R3 : GOSUB 2290 : R3=RT
1590 IF T=1 THEN RI=R1 ELSE RI=R3
1600 GOTO 1730
1610 RT=RI : GOSUB 2290 : RI=RT
1620 IF T=2 THEN 1680
1630 R1=RI : R2=V*R1
1640 RT=R2 : GOSUB 2290 : R2=RT
1650 R3=R1*R2/(R1+R2)
1660 RT=R3 : GOSUB 2290 : R3=RT
1670 GOTO 1730
1680 R3=RI
1690 R2=R3*V
1700 RT=R2 : GOSUB 2290 : R2=RT
1710 R1=R2/(V-1)
1720 RT=R1 : GOSUB 2290 : R1=RT
1730 REM *** CALCULATE CAPACITANCE (C) ***
1740 C1=0
1750 IF F1>0 THEN C1=1/(6.4*F1*R1) * 1E+12
1760 REM *** CALCULATE SLEW RATE, BAND WIDTH GAIN, COMPENSATION CAP ***
1770 SR=6.28*1.414*VM*F2*.000001
1780 BW=10*V*F2
1790 CP=(R1*30)/(R1+R2)
1800 REM *** CALCULATE ACTUAL VOLTAGE GAIN ***
1810 IF T=1 THEN VA=-(R2/R1) ELSE VA=R2/R1+1
1820 REM *** OUPUT RESULTS ***
1830 GOSUB 2580
```

```

1840 PRINT SPC(17); G; "dB ";
1850 IF T=2 THEN PRINT "Non-";
1860 PRINT "Inverting Amplifier"
1870 PRINT
1880 IF C1=0 THEN 1900
1890 CT=C1: GOSUB 2400: C1=CT: C1$=CT$
1900 CT=CP: GOSUB 2400: CP=CT: CP$=CT$
1910 IF T=2 THEN 2060
1920 PRINT "                R2"
1930 PRINT "                +---/\//---+"; SPC(15); "R1=";R1;"OHMS"
1940 PRINT "                |  | \  |"; SPC(15); "R2=";R2;"OHMS"
1950 PRINT "                C    R1  |  | \  |"; SPC(15); "R3=";R3;"OHMS"
1960 PRINT " In >--I(---/\//---+---!- \  |"; SPC(15); " C=";
1970 IF C1=0 THEN PRINT " SHORT" ELSE PRINT C1; C1$
1980 PRINT "                |  \  |"
1990 PRINT "                |  >+----> Out"
2000 PRINT "                R3    |  /"; SPC(18); "Ccomp="; CP; CP$
2010 PRINT "                +---/\//---+ /"; SPC(19); "Ri="; RI; "OHMS"
2020 PRINT "                |  |  | /"; SPC(20); "GAIN (Actual)= x";VA
2030 PRINT "                ---  |/"
2040 PRINT "                -"
2050 GOTO 2180
2060 PRINT "                |\"; SPC(20); "R1="; R1; "OHMS"
2070 PRINT "                C    |\"; SPC(19); "R2="; R2; "OHMS"
2080 PRINT " In >--I(-----+---!+ \"; SPC(18); "R3="; R3; "OHMS"
2090 PRINT "                R3    |  | \"; SPC(18); "C=";
2100 IF C1=0 THEN PRINT " SHORT" ELSE PRINT C1; C1$
2110 PRINT "                +---/\//---+ |  >+----> Out"
2120 PRINT "                |  R1    |  /  |"; SPC(12); "Ccomp="; CP; CP$
2130 PRINT "                +---/\//---+---!- /  |"; SPC(12); "Ri="; RI; "OHMS"
2140 PRINT "                |  |  | /  |"; SPC(12); "GAIN (actual)= x";VA
2150 PRINT "                |  |  | /  |"
2160 PRINT "                ---  |  R2  |"
2170 PRINT "                -  +----/\//---+"
2180 PRINT : PRINT
2190 IF VM<>0 THEN PRINT "Vmax="; VM; "Vrms"
2200 IF F1<>0 THEN PRINT "Fmin="; F1; "Hz"
2210 IF F2<>0 THEN PRINT "Fmax="; F2; "Hz"
2220 IF F2<>0 AND VM<>0 THEN PRINT "Slew Rate (SR)="; SR; "V/uS"
2230 IF F2<>0 THEN PRINT "BAND WIDTH GAIN (BWG)=";BW
2240 IF R2>910001! THEN PRINT "**** WARNING *** R2 values over 1M ohms lead to instability."
2250 PRINT : PRINT "ANOTHER DESIGN (Y/N)?";
2260 A$=INKEY$:IF A$="Y" THEN 1140

```

Designing With Op Amps

```
2270 IF A$<>"N" THEN 2260
2280 STOP
2290 REM *** FIND RESISTOR VALUE ***
2300 REM *** ON ENTRY 'RT' HAS VALUE LOOKING FOR COMMON RESISTANCE ***
2310 REM *** ON EXIT 'RT' HAS COMMON VALUE RESISTANCE ***
2320 A=1 : B=168
2330 C=INT((A+B)/2)
2340 IF C=A THEN 2370
2350 IF Z(C)>RT THEN B=C ELSE A=C
2360 GOTO 2330
2370 IF ABS(Z(C)-RT) > ABS(Z(C+1)-RT) THEN C=C+1
2380 RT=Z(C)
2390 RETURN
2400 REM *** FIND CAPACITOR VALUE ***
2410 REM *** ON ENTRY 'CT' HAS VALUE IN pF LOOKING FOR COMMON CAPACITANCE ***
2420 REM *** ON EXIT 'CT' HAS CAPACITOR VALUE ***
2430 REM *** 'CT$' HAS UNIT ***
2440 CT=INT(CT*10+.5)/10
2450 IF CT<=820 THEN CT$="pF" : GOTO 2490
2460 CT=INT(CT/100+.5)/10
2470 IF CT<=820 THEN CT$="nF" : GOTO 2490
2480 CT=INT(CT/100+.5)/10 : CT$="uF"
2490 A=1 : B=33
2500 C=INT((A+B)/2)
2510 IF C=A THEN 2540
2520 IF Y(C)>CT THEN B=C ELSE A=C
2530 GOTO 2500
2540 IF CT>Y(C) THEN C=C+1
2550 CT=Y(C)
2560 RETURN
2570 REM *** CLEAR SCREEN ***
2575 REM *** PRINT 25 LINES DOWN ***
2580 FOR A=1 TO 25 : PRINT : NEXT A
2585 REM *** PRINT 25 LINES UP ***
2590 FOR A=1 TO 25 : PRINT CHR$(11); : NEXT A
2600 RETURN
2610 REM *** STANDARD RESISTOR TABLE ***
2620 DATA 1, 1.1, 1.2, 1.3, 1.5, 1.6
2630 DATA 1.8, 2, 2.2, 2.4, 2.7, 3
2640 DATA 3.3, 3.6, 3.9, 4.3, 4.7, 5.1
2650 DATA 5.6, 6.2, 6.8, 7.5, 8.2, 9.1
2660 REM *** STANDARD CAPACITOR TABLE ***
2670 DATA 1, 1.5, 1.8, 2.2, 2.7, 3.3, 3.9, 4.7, 5.6, 6.8, 8.2
```